

A COMPUTER PROGRAM FOR PROCESSING ENZYME KINETIC DATA BY THE MAXIMUM LIKELIHOOD METHOD

Ülo VAHER

Eesti Teaduste Akadeemia Eksperimentaalbioloogia Instituut (Institute of Experimental Biology, Estonian Academy of Sciences), Instituudi tee 11, Harku, EE-3051 Harjumaa, Eesti (Estonia)

Presented by A. Aaviksaar

Received 12 January 1995, accepted 9 February 1995

Abstract. A personal computer complex program written in BASIC for the estimation of the Michaelis–Menten enzyme kinetics and inhibition constants or ligand binding process parameters and their confidence limits by the maximum likelihood method is presented. The program needs no provisional estimates of the constants for starting and is free from converging problems. To compare the method and the program with other methods (nonlinear regression by weighted least squares method and nonparametric “direct linear plot” developed by Cornish-Bowden), large-scale computer simulation experiments by the Monte Carlo method were performed. The results of these experiments demonstrate that the program works faster and provides the parameter estimates with less bias or variance. The listing of the program is presented.

Key words: BASIC, computer program, enzyme kinetic constants, Michaelis–Menten equation, enzyme inhibition, ligand binding, maximum likelihood method, Monte Carlo method.

INTRODUCTION

In biochemistry one of the fundamental equations for the description of enzymatic reactions is the Michaelis–Menten equation

$$v = \frac{V_{\max} \cdot s}{K_M + s}, \quad (1)$$

where v is the reaction initial velocity, s is the substrate concentration, and K_M and V_{\max} are constant parameters (K_M is the Michaelis constant and V_{\max} is the reaction maximal initial velocity). Various theoretical and

practical aspects of developing valid methods for estimating the parameters (constants) of the equation and their errors have been discussed in literature [1-37]. The constants K_M and V_{\max} are estimated on the basis of experimental data (s_i, v_i) , the substrate concentration, and the corresponding reaction initial rate, where $i = 1, 2, \dots, n$ is the number of experimental points.

The existing methods differ in the underlying mathematical principles, the amount of computational work needed, and the quality of the estimation (its accuracy and precision). In the very beginning of the computer era several programs were developed for calculating K_M and V_{\max} on the mainframes in computer centres [9, 19, 20, 23, 32, 33, 36]. However, these programs have not been used widely because their application is accompanied with certain inconveniences. In the last decades almost all biochemical laboratories have been equipped with personal microcomputers. As a result, the need in efficient and reliable personal computer programs has essentially increased and several microcomputer programs have been published [8, 14, 15, 18, 24, 25]. These programs differ in the underlying statistical assumptions and computational algorithms and therefore in efficiency and reliability. They may give unreliable (biased or imprecise) estimates of the parameters and their variances depending on the experimental conditions used. In order to evaluate the existing computational methods and published programs their theoretical statistical background has to be considered.

THEORETICAL BACKGROUND

The statistical methods can be divided into two categories: (1) parametric ones, which assume that the shape and breadth of the experimental error distribution is known, and (2) nonparametric or distribution-free ones, which make fewer (but still some) assumptions about the form of the distributions. The parametric methods are usually more efficient (in the sense that they produce more precise and accurate results) because of the extra information on the error distribution they use; on the other hand, if this information is incorrect they may give fallacious answers [4, 5]. Thus, to select an appropriate statistical method one has to know about the random error in one's data. Most special investigations have shown that the error distribution in enzyme kinetic measurements is Gaussian and has a constant coefficient of variation [1, 34].

In the context of curves fitting (including straight lines), the least square method requires a Gaussian distribution of errors in the dependent variable (in v in the Michaelis-Menten equation). This is the main reason why all the linear regression methods (and accordingly the computer programs) using least square methods are inadequate because through linearization transformations they also introduce transformations into the error distribution which loses its originally normal distribution and constant relative magnitude.

In the parametric methods the method of least squares predominates. Using this method such estimates of fitted curves can be calculated that the sum of the squares of the deviates of the experimental points from the fitted curve is minimal. If the error distribution is heteroscedastic, the correcting weight factors must be included into the least square expressions. This complicates the calculations. But the most apparent disadvantage of the nonlinear regression by the least square method is that it needs some initial estimates for unknown parameters and the iterative computing process begins to diverge if these initial estimates are not close enough to their exact values.

The best estimators for any statistical parameter are those which are unbiased (accurate), that is, their expectation (mean value) is equal to the true value, and their variation is minimal (the estimate is precise). Such estimators can be derived by the maximum likelihood method which finds such values for the unknown parameters of a given distribution by which the probability of obtaining the values observed in an experiment is maximal.

Michaelis–Menten equation. If the experimental error of the Michaelis–Menten-type enzyme reaction velocity v is proportional to v and it has a normal distribution, then the maximum likelihood function is:

$$L = \prod_{i=1}^n \frac{1}{\sqrt{2\pi}\sigma_i} \exp \left\{ -\frac{(v_i - v)^2}{2\sigma_i^2} \right\}, \quad (2)$$

where

$$v = \frac{V_{\max} \cdot s_i}{K_M + s_i} \quad (3)$$

and

$$\sigma_i = \frac{s_i}{K_M + s_i} \cdot \sigma_{0i}, \quad \sigma_{0i} \sim N(0, \sigma_0^2). \quad (4)$$

In order to find the maximum likelihood estimates for parameters V_{\max} and K_M , we have to differentiate the likelihood function L (or more conveniently the logarithm of L) against the parameters, set the derivatives equal to 0, and solve Eqs. (5), (6), and (7) simultaneously.

$$\frac{\partial \ln(L)}{\partial \sigma_0^2} = -\frac{n}{2\sigma_0^2} + \frac{1}{2(\sigma_0^2)^2} \sum_{i=1}^n (v_i + K_M v_i / s_i - V_{\max})^2 = 0, \quad (5)$$

$$\frac{\partial \ln(L)}{\partial V_{\max}} = \frac{1}{2\sigma_0^2} \sum_{i=1}^n (v_i + K_M v_i / s_i - V_{\max}) = 0, \quad (6)$$

$$\frac{\partial \ln(L)}{\partial K_M} = \sum_{i=1}^n \frac{s_i}{s_i + K_M} - \frac{1}{\sigma_0^2} \sum_{i=1}^n (v_i + K_M v_i / s_i - V_{\max}) = 0. \quad (7)$$

The above equations may be rewritten in a more convenient form:

$$\sigma_0^2 = \frac{1}{n} (S_3 - 2S_2 K_M + S_1 K_M^2), \quad (8)$$

$$V_{\max} = \frac{1}{n} \sum_{i=1}^n v_i + K_M \sum_{i=1}^n (v_i / s_i), \quad (9)$$

$$(S_1 K_M^2 + 2S_2 K_M + S_3) Q_1 - S_1 K_M - S_2 = 0, \quad (10)$$

where

$$S_1 = \sum_{i=1}^n (v_i / s_i - K_1)^2, \quad S_2 = \sum_{i=1}^n (v_i / s_i - K_1) (v_i - K_3),$$

$$S_3 = \sum_{i=1}^n (v_i - K_3)^2, \quad (11)$$

$$K_1 = \frac{1}{n} \sum_{i=1}^n (v_i / s_i), \quad K_2 = \frac{1}{n} \sum_{i=1}^n (1 / s_i),$$

$$K_3 = \frac{1}{n} \sum_{i=1}^n v_i, \quad (12)$$

and

$$Q_1 = \frac{1}{n} \sum_{i=1}^n 1 / (K_M + s_i). \quad (13)$$

Eq. (10) for K_M is nonlinear, so K_M cannot be expressed explicitly from it. In our program we use the Newton-Raphson method to calculate K_M by iteration:

$$K_M = K_M^0 - \frac{F(K_M^0)}{F'(K_M^0)}, \quad (14)$$

where $F(K_M)$ is the function in (10) and $F'(K_M)$ is its derivative:

$$F(K_M^0) = S_2 + S_1 K_M^0 - (S_3 + 2S_2 K_M^0 + S_1 K_M^0) Q_1, \quad (15)$$

$$F'(K_M^0) = S_1 + (S_3 + 2S_2K_M^0 + S_1(K_M^0)^2)Q_2 - 2(S_2 + S_1K_M^0)Q_1, \quad (16)$$

and

$$Q_2 = \frac{1}{n} \sum_{i=1}^n 1/(K_M^0 + s_i)^2. \quad (17)$$

A very good initial approximation for K_M is

$$K_M^0 = \frac{K_1S_3 - K_3S_2}{K_3S_1 - K_1S_2}. \quad (18)$$

In most cases only one iteration is sufficient to calculate K_M with reasonable precision (relative error less than 0.0005).

The attractive features of the maximum likelihood method include also the possibility to calculate the large sample approximations for variances of the parameters K_M and V_{\max} and their covariance by inverting the matrix of second-order derivatives of logarithmic likelihood function, $\ln L$

$$\begin{bmatrix} \text{var}(K_M) & \text{cov}(K_M, V_{\max}) \\ \text{cov}(K_M, V_{\max}) & \text{var}(V_{\max}) \end{bmatrix} = \frac{1}{D} \begin{bmatrix} \frac{\partial^2 \ln(L)}{\partial V_{\max} \partial V_{\max}} & \frac{\partial^2 \ln(L)}{\partial K_M \partial K_M} \\ \frac{\partial^2 \ln(L)}{\partial K_M \partial V_{\max}} & \frac{\partial^2 \ln(L)}{\partial K_M \partial K_M} \end{bmatrix}, \quad (19)$$

where D is the determinant of the matrix of the second derivatives of $\ln L$. After differentiation of $\ln L$ and inverting the matrix we arrive at the following result:

$$\text{var}(K_M) = \frac{\sigma_0^2}{\left(1 + \frac{2\sigma_0^2}{V_{\max}^2}\right) \sum_{i=1}^n \left(\frac{V_{\max}}{K_M + s_i} - \frac{1}{n} \sum_{i=1}^n \frac{V_{\max}}{K_M + s_i}\right)^2}, \quad (20)$$

$$\text{var}(V_{\max}) = \frac{\text{var}(K_M) + (1/n) \left(\sum_{i=1}^n V_{\max}/(K_M + s_i)\right)^2}{n}, \quad (21)$$

$$\text{cov}(K_M, V_{\max}) = \frac{\text{var}(K_M) \sum_{i=1}^n V_{\max}/(K_M + s_i)}{n}. \quad (22)$$

Determination of the inhibition constants. When a competitive or noncompetitive inhibitor is present then the rate equations can be described respectively by (23) and (24) :

$$v = \frac{V_{\max} s}{K_M (1 + I/K_i) + s}, \quad (23)$$

$$v = \frac{V_{\max} s}{(1 + I/K_i) (K_M + s)}. \quad (24)$$

The inhibition constants are determined in two steps: at first the "apparent" Michaelis-Menten equation parameters K_M^a and V_{\max}^a are calculated by the main subroutine and then the K_i values for competitive and noncompetitive inhibition can be calculated respectively by formulas (25) and (26):

$$K_i = \frac{K_M I}{K_M^a - K_M}, \quad (25)$$

$$K_i = \frac{V_{\max} I}{V_{\max} - V_{\max}^a}. \quad (26)$$

Determination of the ligand binding constants in general. The ligand binding process is often described by the following equation:

$$B = \frac{B_{\max} s}{K_B + s}, \quad (27)$$

where s is free ligand concentration, B is bound ligand concentration, and K_B is the binding constant. The difficulty involved in ligand binding experiments is that the experimentally controlled error-free parameter is not the free ligand concentration but the total (free + bound) ligand. So the parameters K_B and B_{\max} can be calculated only by an iterative process. Starting with some provisional estimates K_M^0 and B_{\max}^0 for the parameters, the free ligand concentration can be estimated on the basis of Eq. (27), which becomes quadratic in respect to s :

$$s = \frac{T - K_B - B_{\max}}{2} + \sqrt{\frac{(T - K_B - B_{\max})^2}{4} + TB_{\max}}. \quad (28)$$

Using these values for s we can get improved estimates for K_B and B_{\max} by the basic subroutine. This iterative process will be continued until the values for K_B and B_{\max} stabilize.

The integrated Michaelis–Menten equation. Substituting in Eq. (1) $v = -\frac{ds}{dt}$ and integrating by t we get the integrated Michaelis–Menten equation (sometimes also called a “progress curve”):

$$t = \frac{(s_0 - s) + K_M \ln(s_0/s)}{V_{\max}}, \quad (29)$$

where s is the substrate concentration at time moment t . Alternatively, when the product concentration is measured instead the substrate in the course of the experiment, the integrated equation will be

$$t = \frac{p - K_M \ln(1 - (p/s_0))}{V_{\max}}, \quad (30)$$

where s_0 is the initial substrate concentration. If we draw a chord joining two points (t_1, s_1) and (t_2, s_2) on the progress curve of the substrate concentration against time then its slope is $(s_1 - s_2)/(t_1 - t_2)$ and this equals the rate (i.e. the slope of the tangent) at some intermediate substrate concentration s_c , given by:

$$s_c = \frac{s_1 - s_2}{\ln(s_1/s_2)}. \quad (31)$$

The corresponding rate is:

$$v_c = \frac{s_1 - s_2}{t_1 - t_2}. \quad (32)$$

So, in principle each pair of points on the progress curve enables to calculate a point (s_c, v_c) on the traditional Michaelis–Menten curve and finally the constants K_M and V_{\max} if the experimental errors in s_c were negligible. In general they may be used only for calculating provisional estimates of the kinetic constants. The points on the progress curve must be used for calculating with maximum precision the initial rate v_0 , corresponding to s_0 , using Eq. (33):

$$v_0 = \{s_0 \ln(s_0/s_1) [(s_0 - s_2) \ln(s_0/s_1) - (s_0 - s_1) \ln(s_0/s_2)]\} / \\ / \{t_1 [(s_0 - s_2) \ln(s_0/s_1) - (s_0 - s_1) \ln(s_0/s_2)] + \\ + [t_2 \ln(s_0/s_1) - t_1 \ln(s_0/s_2)] [s_0 \ln(s_0/s_1) - (s_0 - s_1)]\}. \quad (33)$$

An analogous formula can be derived for experiments when the product is recorded substituting p_i for $(s_0 - s_i)$ and $\ln(1 - p_i/s_0)$ for $\ln(s_0/s_i)$, ($i = 1, 2$).

DESCRIPTION AND LISTING OF THE PROGRAM

The program (see Appendix) is written in GWBASIC, a widespread version of programming languages, so it can be implemented on every computer. The requirements for computer resources are minimal: the source code needs only 10 kB of computer memory.

The program is user-friendly: interactive, self-explanatory, and menu-driven. It starts with displaying the main menu prompting to make a choice between five possibilities:

1) for computing the Michaelis–Menten equation parameters, their standard errors and 95%-confidence limits. It has to be selected when only initial concentrations and velocities are provided;

2) for calculating the Michaelis–Menten equation parameters from progress curves (it may be used when there is no substrate inhibition or enzyme inactivation in the course of reaction);

3) for calculating the inhibition constants when inhibition is competitive;

4) for calculating the inhibition constants when inhibition is non-competitive;

5) for calculating the general ligand binding (or saturation) constants.

The calculations are executed by common and specific subroutines. It is worth mentioning that the problem of getting the exact initial rate is often overlooked. Namely, it is the slope of the tangent to the progress curve in the starting point but in practice it is often substituted by the slope of a chord connecting the starting point and the first point on the progress curve, which is always less than the slope of the tangent. A special subroutine in the program (lines 4000–4170) provides exact values for initial rates by calculating slopes of tangents using Eq. (27). Besides calculating the parameters, their errors and confidence limits, the program also computes the theoretical values of rates and the differences between experimental and theoretical values. The program includes also a graphic output subroutine (lines 2000–2185) for displaying the calculated theoretical kinetic curve and the experimental points on it.

In the program the arrays for experimental data $S()$, $V()$, and $T()$ are dimensioned for up to 50 points. If the number of experimental points exceeds this value, the arrays must be redimensioned up to the appropriate value in line 32.

COMPUTER EXPERIMENTS

In order to evaluate the maximum likelihood method estimates for Michaelis–Menten-type equation parameters and our program for their calculation, large-scale computer experiments were designed and performed by the Monte Carlo method. The values for concentration were $s = 1, 3, 9, 27, 81, 243$ (arbitrary units). The corresponding “correct” rate values were calculated by formula (1) using the values $K_M = 10$ and

$V_{\max} = 25$, and then random "experimental errors" were added. These errors were generated by computer and they had a normal distribution with zero mean and with the standard deviation proportional to the "correct" value. The normally distributed random numbers were calculated using formulas (34).

$$R_1^i = \cos(2\pi X_{i-1}) \sqrt{2 \ln(1/X_i)},$$

$$R_2^i = \sin(2\pi X_{i-1}) \sqrt{2 \ln(1/X_i)}, \quad (34)$$

where X_i and X_{i-1} are two random numbers with an equal distribution in the range (0, 1). They were generated by the BASIC internal generator. The distribution of the random numbers was controlled in a special experiment with 120 000 generated numbers. Their distribution is presented in Fig. 1 indicating that it was really normal. In order to examine the influence of the error magnitude on the estimators' goodness, five series of experiments were carried out with different coefficients of the variation of the experimental error (5, 10, 15, 20, and 25%). In each series 10 000 experimental curves were generated and the Michaelis-Menten equation parameters K_M and V_{\max} were calculated by three methods: (1) our program, (2) the weighted nonlinear regression using the least squares method, and (3) the nonparametric method proposed by Cornish-Bowden [11-13]. The summarized results are presented in the Table. They demonstrate that the magnitude of the errors (their coefficient of variation) has a very strong effect on the estimate quality. So, when the random error is small (coefficient of variation $v \leq 5\%$), all the studied methods will give fairly good and similar estimates with a bias less than 0.5%. A five-fold rise in the error magnitude ($v = 25\%$) will result in up to




Fig. 1. Distribution of computer-simulated 120 000 random numbers used for the simulation of the "errors" in the computer experiments. The dotted line represents the theoretical normal distribution curve.

20 times increase in the K_M estimate's bias if the traditional least squares method is used. The maximum likelihood method provided less biased estimates than the other methods by all the experimental error levels used. The nonparametric method gave estimates with a minimal bias but their variances were essentially greater than by our method, which causes also greater mean square errors of the estimates.

The results of computer-simulated experiments in comparison with different methods for the estimation of Michaelis–Menten equation parameters

Errors of the estimates	Coefficient of error variation, %	Method of estimation of the parameters		
		Maximum likelihood	Least squares	Nonparametric
ΔK_M	5	0.024	-0.048	0.019
	10	0.086	-0.189	0.006
	15	0.195	-0.423	-0.053
	20	0.345	-0.753	-0.179
	25	0.536	-1.173	-0.417
s_K	5	0.628	0.625	0.681
	10	1.264	1.239	1.365
	15	1.921	1.840	2.060
	20	2.610	2.409	2.746
	25	3.322	2.950	3.416
ΔV_{\max}	5	0.013	-0.062	0.011
	10	0.053	-0.245	-0.009
	15	0.119	-0.547	-0.091
	20	0.208	-0.955	-0.252
	25	0.319	-1.470	-0.535
s_V	5	0.847	0.846	1.009
	10	1.692	1.686	2.016
	15	2.536	2.513	3.022
	20	3.364	3.315	4.028
	25	4.196	4.099	5.041
$MSE(K_M)$	5	0.628	0.627	0.681
	10	1.267	1.253	1.365
	15	1.931	1.888	2.061
	20	2.633	2.524	2.752
	25	3.365	3.175	3.441
$MSE(V_{\max})$	5	0.847	0.848	1.009
	10	1.693	1.847	2.016
	15	2.539	2.572	3.023
	20	3.370	3.450	4.036
	25	4.208	4.355	5.069

ΔK_M – bias of K_M , s_K – standard deviation of K_M , s_V – standard deviation of V_{\max} , ΔV_{\max} – bias of V_{\max} , MSE – mean square error. For the calculation of each estimate and their errors 10 000 Michaelis–Menten curves (each containing 6 points) were simulated by computer. The values for constants and substrates were (in arbitrary units): $K_M = 10$; $V_{\max} = 25$; $s_i = 1, 3, 9, 27, 81, 243$.

The statistical distribution of the estimates of K_M and V_{\max} was approximately normal as predicted from the theory underlying the maximum likelihood method (see Figs. 1, 2). This was also checked by the χ^2 -test ($P > 0.2$) and Kolmogorov–Smirnov test ($P > 0.1$).

So, in conclusion, the computer-simulated experiments proved the superiority of the maximum likelihood method and the corresponding computer program. The advantages of this method are even more pronounced when the number of experimental points increases [30].




Fig. 2. Distribution of 10 000 Michaelis–Menten equation parameters calculated by the proposed program in computer-simulated experiments. The number of points on each curve is 6; the substrate concentrations, s_i , are 1, 3, 9, 27, 81, and 243 (arbitrary units). The dotted line – normal distribution curve. Above: K_M (real value, 10); below: V_{\max} (real value, 25).

ACKNOWLEDGEMENT

The author is grateful to Professor A. Aaviksaar for his help in preparing the manuscript.

APPENDIX

```

10 REM MICHAEL. A program for estimating enzyme kinetic constants
15 REM and fitting curves in enzyme kinetic studies
20 REM Author: Ylo Vaher.
25 REM Institute of Experimental Biology
30 REM *****
32 DIM S(50),V(50),T(50),X(50)
35 CLS
40 PRINT "
45 PRINT " MICHAEL
50 PRINT " This program estimates constants and fits curves
55 PRINT " in enzyme kinetic studies"
60 PRINT
65 PRINT
70 PRINT " *****"
75 PRINT " * *"
80 PRINT " * MAIN MENU *"
85 PRINT " * *"
90 PRINT " * *"
95 PRINT " * MICHAELIS-MENTEN equation (ordinary).....1 *"
100 PRINT " * MICHAELIS-MENTEN equation (integrated).....2 *"
105 PRINT " * ENZYME INHIBITION (competitive).....3 *"
110 PRINT " * ENZYME INHIBITION (noncompetitive).....4 *"
115 PRINT " * LIGAND BINDING.....5 *"
120 PRINT " * EXIT.....6 *"
125 PRINT " * *"
130 PRINT " *****"
135 PRINT " Enter 1,2,3,4,5 OR 6":INPUT IM
140 IF IM=1 OR IM=2 OR IM=3 OR IM=4 OR IM=5 OR IM=6 THEN 145 ELSE 135
145 ON IM GOTO 150,315,195,260,555,999
150 GOSUB 3000:GOSUB 4000:IF G=1 THEN GOSUB 1000:GOTO 160
155 GOSUB 3200
160 CLS:PRINT TAB(30);"COMPUTING RESULTS:"
165 PRINT "Km=";B:PRINT "St.error of Km=";P;
170 PRINT "95%-confidence limits (";B-1.96*P;B+1.96*P;)"
175 PRINT "Vmax=";A:PRINT "St.error of Vmax=";Q;
180 PRINT "95%-confidence limits (";A-1.96*Q;A+1.96*Q;)"
185 GOSUB 3600
190 GOTO 35
195 GOSUB 3100:GOSUB 4000:IF G=1 THEN GOSUB 1000: GOTO 205
200 GOSUB 3200
205 INPUT "Michaelis constant Km=";KM
210 INPUT "St. error of Km=";SM
215 INPUT "Inhibitor concentration=";IC
220 Y=B:X=KM:SY=VB:SX=SM*SM
225 KI=X*IC/(Y-X)
230 S=IC/(Y-X)*(Y-X)*SQR(Y*Y*SX+X*X*SY)
235 PRINT TAB(30);"COMPUTING RESULTS:"
240 PRINT "Inhibition constant Ki=";KI
245 PRINT "St.error of Ki=";S;"confidence limits (";KI-1.96*S;KI+1.96*S;)"
250 GOSUB 3600
255 GOTO 35
260 GOSUB 3400:GOSUB 4000:IF G=1 THEN GOSUB 1000:GOTO 264
262 GOSUB 3200
264 INPUT "Maximum velocity Vmax=";VM
265 INPUT "St. error of Vmax=";SM
270 INPUT "Inhibitor concentration=";IC
275 Y=A:X=VM:SY=VA:SX=VM*VM
280 KI=X*IC/(X-Y)
285 S=IC/(Y-X)*(Y-X)*SQR(Y*Y*SX+X*X*SY)
290 PRINT TAB(30);"COMPUTING RESULTS:"
295 PRINT "Inhibition constant Ki=";KI
300 PRINT "St. error of Ki=";S;"confidence limits (";KI-1.96*S;KI+1.96*S;)"
305 GOSUB 3600
310 GOTO 35
315 PRINT "Estimating constants of the integrated Michaelis-Menten equation"
320 PRINT " ('progress curve')"
325 PRINT
330 PRINT " (So-S)+Km*ln(So/S) P-Km*ln(P/So)"
335 PRINT " t=----- or t=-----"
340 PRINT " Vmax Vmax"
345 PRINT
350 PRINT "where:"
355 PRINT " Km - Michaelis constant"

```

```

360 PRINT " Vmax- maximum rate"
365 PRINT " So - initial substrate concentration"
370 PRINT " S - substrate concentration at time t"
375 PRINT " P - product concentration at time t"
380 PRINT "*****"
385 PRINT "*" MENU "*"
390 PRINT "*" The measured quantity is SUBSTRATE concentration....1 "*"
395 PRINT "*" -,- PRODUCT -,- ....2 "*"
400 PRINT "*****"
405 INPUT " Enter 1 or 2";IV
410 INPUT "Initial substrate concentration";S0
415 INPUT "The number of measured points on the curve";N
425 ON IV GOTO 430,475
430 FOR I=1 TO N
435 PRINT "point nr.";I;:INPUT "time, substrate concentration";T(I),X(I)
440 NEXT I
445 GOSUB 3350
450 N=N-1;FOR I=1 TO N:
455 S1=X(I):S2=X(I+1)
460 S(I)=(S1-S2)/LOG(S1/S2):V(I)=(S1-S2)/(T(I+1)-T(I))
465 NEXT I
470 GOTO 515
475 FOR I=1 TO N
480 PRINT "point No.";I;:INPUT "time, product concentration";T(I),X(I)
485 NEXT I
490 GOSUB 3350
495 N=N-1;FOR I=1 TO N
500 P1=X(I):P2=X(I+1)
505 S(I)=(P2-P1)/LOG((S0-P1)/(S0-P2)):V(I)=(P2-P1)/(T(I+1)-T(I))
510 NEXT I
515 GOSUB 1000
520 PRINT TAB(30);"COMPUTING RESULTS:"
525 PRINT "Km=";B;"St.error of Km=";P;
530 PRINT "95%-confidence limits (";B-1.96*P;B+1.96*P;)"
535 PRINT "Vmax=";A;"St.error of Vmax=";Q;
540 PRINT "95%-confidence limits (";A-1.96*Q;A+1.96*Q;)"
545 GOSUB 3600
550 GOTO 35
555 CLS
560 PRINT " LIGAND BINDING "
565 PRINT
570 PRINT " Bmax * s "
575 PRINT " B = ----- "
580 PRINT " K + s "
585 PRINT
590 PRINT "where:
595 PRINT " s - free ligand concentration "
600 PRINT " B - bound ligand concentration "
605 PRINT " K - binding constant "
610 PRINT " Bmax - saturating binding "
612 PRINT:PRINT "*****"
615 INPUT "Number of experimental points=";N
625 FOR I=1 TO N
630 PRINT "Point nr.";I;"ligand concentrations:"
635 INPUT "total=";T(I):INPUT "bound=";V(I)
640 NEXT I
645 PRINT "You have entered:"
650 PRINT "point No.";I;"total","free"
655 FOR I=1 TO N:PRINT I,T(I),V(I):NEXT I
660 INPUT "Are the data entered correct (y/n)";VI$
665 IF VI$="n" THEN GOTO 625
670 INPUT "Starting estimates for K and Bmax=";B,A;E=.01
675 B0=B:A0=A
680 FOR I=1 TO N
685 X=(T(I)-B-A)/2:S(I)=X+SQR(X*X+T(I)*B):PRINT I,S(I)
690 NEXT I
695 GOSUB 1000:PRINT "A=";A;"B=";B:INPUT "continue";J$:IF J$="n" THEN END
700 IF ABS((B-B0)/B0)>E OR ABS((A-A0)/A0)>E THEN GOTO 675
705 PRINT TAB(30);"COMPUTING RESULTS:"
710 PRINT "BINDING CONSTANT K=";B;"ST.ERROR OF K=";P;
715 PRINT "95%-confidence limits(";B-1.96*P;B+1.96*P;)"
720 PRINT "Bmax=";A;"St.error of Bmax=";Q;
725 PRINT "95%-confidence limits (";A-1.96*Q;A+1.96*Q;)"
730 GOSUB 3600:GOTO 35
999 END
1000 XK=0:VK=0:SK=0
1010 FOR I=1 TO N
1020 X=V(I)/S(I):XK=XK+X:VK=VK+V(I):X(I)=X

```

```

1030 NEXT I: XK=XK/N: VK=VK/N
1040 SX=0: SV=0: SXV=0
1050 FOR I=1 TO N
1060 X=X(I)-XK: V=V(I)-VK
1070 SX=SX+X*X: SV=SV+V*V: SXV=SXV+X*V
1080 NEXT I
1090 B=(XK*SV-VK*SXV)/(VK*SX-XK*SXV)
1100 Q1=0: Q2=0
1110 FOR I=1 TO N
1120 X=1/(S(I)+B): Q1=Q1+X: Q2=Q2+X*X
1130 NEXT I: Q1=Q1/N: Q2=Q2/N
1140 SL=SV+2*SXV*B+SX*B*B
1150 F=SXV+SX*B-SL*Q1: F1=SX+SL*Q2-2*(SXV+SX*B)*Q1
1160 B=B-F/F1: IF ABS(F/F1/B)>.00005 GOTO 1100
1170 REM
1180 A=VK+B*XK
1190 S2=(SV+2*B*SXV+B*B*SX)/(N-2)
1200 UK=0
1210 FOR I=1 TO N
1220 UK=UK+A/(B+S(I))
1230 NEXT I: UK=UK/N
1240 D=0
1250 FOR I=1 TO N
1260 D=D+(A/(S(I)+B)-UK)^2
1270 NEXT I
1280 VB=S2/((2*S2/A^2+1)*D)
1290 VA=S2/N+UK*UK*VB
1300 COV=UK*VB
1310 SQ=0
1320 FOR I=1 TO N
1330 X=A*S(I)/(S(I)+B): U(I)=X
1340 SQ=SQ+(V(I)-X)^2
1350 NEXT I
1360 P=SQR(VB): Q=SQR(VA)
1370 RETURN

2000 REM GRAPHIC *****
2010 PRINT "Terminal points of y-axis (Ymin,Ymax)": INPUT YMIN,YMAX
2020 PRINT "Terminal points of x-axis (Xmin,Xmax)": INPUT XMIN,XMAX
2030 YM=YMAX-YMIN: YN=INT(YM/YD+.1): DEF FNY(X)=(X-YMIN)/YM
2040 XM=XMAX-XMIN: XN=INT(XM/XD+.1): DEF FNX(X)=(X-XMIN)/XM
2050 SCREEN 9: KEY OFF: CLS
2060 VIEW (150,20)-(610,310)
2070 WINDOW (-.1,-.1)-(1,1)
2080 LINE (0,0)-(0,1): LINE (0,0)-(1,0): LINE (1,0)-(1,1): LINE (1,1)-(0,1)
2090 FOR I=1 TO 9: Y=I*.1: LINE (0,Y)-(.01,Y): NEXT I
2100 K=XD/XM: FOR I=1 TO 9: X=I*.1: LINE (X,0)-(X,-.02): NEXT I
2110 FOR I=1 TO N: XI=S(I): YI=V(I): X=FNX(XI): Y=FNY(YI): CIRCLE(X,Y),.01: NEXT I
2120 NP=100: DX=XM/NP: FOR I=1 TO NP: XI=XMIN+I*DX: X=FNX(XI)
2130 YI=A*XI/(B+XI): Y=FNY(YI): PSET(X,Y): NEXT I
2140 LOCATE 2,20: PRINT YMAX: LOCATE 21,20: PRINT YMIN
2150 LOCATE 22,24: PRINT XMIN: LOCATE 22,75: PRINT XMAX
2160 LOCATE 2,10: PRINT "Rate": LOCATE 23,45: PRINT "concentration"
2170 LOCATE 10,5: PRINT "Km=": KM: LOCATE 12,5: PRINT "Vmax=": VM
2180 LOCATE 24,30: INPUT "continue": V$
2185 CLS 0: SCREEN 0
2190 RETURN: REM *****
3000 CLS
3010 PRINT "Estimating constants of the Michaelis-Menten equation "
3020 PRINT
3030 PRINT "          Vmax * s"
3040 PRINT " v =-----"
3050 PRINT "          Km + s"
3060 PRINT "where:"
3070 PRINT " Km - Michaelis constant"
3080 PRINT " Vmax - maximum rate"
3090 PRINT " s - substrate concentration"
3092 PRINT " v - reaction rate "
3094 PRINT "*****"
3096 RETURN
3100 CLS
3105 PRINT "Estimating inhibition constants (competitive inhibition)"
3110 PRINT
3115 PRINT "          Vmax * s"
3120 PRINT " v =-----"
3125 PRINT "          Km*(1+I/Ki)+ s"
3130 PRINT "where:"
3135 PRINT " Ki - inhibition constant"
3140 PRINT " Km - Michaelis constant"

```

```

3145 PRINT"          Vmax - maximum rate"
3150 PRINT"          s - substrate concentration"
3155 PRINT"          v - reaction rate"
3160 PRINT
3170 RETURN
3200 PRINT
3210 PRINT "Enter number of experimental points":INPUT N
3230 FOR I=1 TO N
3240 PRINT "point No. ";I;:PRINT "concentration, rate";:INPUT S(I),V(I)
3250 NEXT I
3260 PRINT "You have entered:"
3270 PRINT "point No.", "concentration", "rate"
3280 FOR I=1 TO N:PRINT I,S(I),V(I):NEXT I
3290 INPUT "Are the entered data correct (y/n)";VI$
3300 IF VI$="n" THEN GOTO 3240
3310 GOSUB 1000
3320 RETURN
3350 PRINT "You have entered:"
3370 FOR I=1 TO N:PRINT I,T(I),X(I):NEXT I
3380 INPUT "Are the entered data correct (y/n)";VI$
3390 IF VI$="n" THEN GOTO 3350
3395 RETURN
3400 CLS
3410 PRINT "Estimating inhibition constants (noncompetitive inhibition)"
3420 PRINT
3430 PRINT "          Vmax * s"
3440 PRINT "          v -----"
3450 PRINT "          (1+I/Ki)(Km+s)"
3460 PRINT "where:"
3470 PRINT "          Ki - inhibition constant"
3480 PRINT"          Km - Michaelis constant"
3490 PRINT"          Vmax - maximum rate"
3500 PRINT"          s - substrate concentration"
3510 PRINT"          v - reaction rate"
3520 PRINT
3530 RETURN
3600 PRINT "Point conc.      rate          computed          diff. diff.(%)"
3605 PRINT "-----"
3610 FOR I=1 TO N
3620 V=V(I):Y=A*S(I)/(B+S(I)):X=V-Y:Z=INT(X/Y*10000)/100
3630 PRINT I;TAB(5);S(I);TAB(15);V;TAB(30);Y;TAB(45);X;TAB(60);Z
3640 NEXT I
3650 INPUT "Graphic(y/n)";IV$:IF IV$="y" THEN GOSUB 2000
3660 RETURN
4000 REM *****
4010 PRINT "Do you wish to calculate EXACT INITIAL RATES by using"
4020 PRINT "two first points on progress curve (y/n)":INPUT V$
4030 IF V$<>"y" THEN RETURN
4040 INPUT "number of experiments (substrate concentrations)";N
4050 FOR I=1 TO N
4060 PRINT "Experiment nr. ";I
4070 INPUT "initial substrate concentration=";S0:S(I)=S0
4080 PRINT "First point on the progress curve"
4090 INPUT "Time and substrate concentration respectively (t1,s1)=";T1,S1
4100 PRINT "Second point on the progress curve"
4110 INPUT "Time and substrate concentration respectively (t2,s2)=";T2,S2
4120 L1=LOG(S0/S1):L2=LOG(S0/S2):S1=S0-S1:S2=S0-S2
4130 A=(T2*L1-T1*L2)/(S2*L1-S1*L2)
4140 V0=S0*L1/(T1+A*(S0*L1-S1)):V(I)=V0
4150 NEXT I
4160 G=1
4170 RETURN

```

REFERENCES

1. Askelöf, P., Korsfeldt, M., Mannervik, B. Error structure of enzyme kinetic experiments. Implication for weighting in regression analysis of experimental data. - Eur. J. Biochem., 1976, 69, 1, 61-67.
2. Atkins, G. L. A versatile digital computer program for nonlinear regression analysis. - Biochim. Biophys. Acta, 1971, 252, 3, 405-420.

3. Atkins, G. L. A simple digital computer program for estimating the parameters of the Hill equation. – *Eur. J. Biochem.*, 1973, **33**, 175–180.
4. Atkins, G. L., Nimmo, I. A. A comparison of seven methods for fitting Michaelis–Menten equation. – *Biochem. J.*, 1975, **149**, 3, 775–777.
5. Atkins, G. L. Current trends in the estimation of Michaelis–Menten parameters. – *Anal. Biochem.*, 1980, **104**, 1, 1–9.
6. Baldwin, R. L., Thornley, J. H., Breever, D. E. Metabolism of the lactating cow. II. Digestive elements of a mechanistic model. – *J. Dairy. Res.*, 1978, **54**, 1, 107–131.
7. Brooks, S. P., Suelter, C. R. Estimating enzyme kinetic parameters: A computer program for linear regression and nonparametric analysis. – *Int. J. Biomed. Comput.*, 1986, **19**, 2, 89–99.
8. Brady, J. F. A BASIC program for the estimation of Michaelis–Menten parameters by the direct linear plot. – *Comput. Meth. Progr. Biomed.*, 1989, **28**, 4, 271–272.
9. Cleland, W. W. Computer programmes for processing enzyme kinetic data. – *Nature*, 1963, **4879**, 463–465.
10. Colquhoun, D. A comparison of estimators for a two-parameter hyperbola. – *J. Royal Statist. Soc. Series C*, 1969, **18**, 2, 130–140
11. Cornish-Bowden, A., Eisenthal, R. Estimation of Michaelis constant and maximum velocity from the direct linear plot. – *Biochim. Biophys. Acta*, 1978, **53**, 1, 268–272.
12. Cornish-Bowden, A., Endrenyi, L. Robust regression of enzyme kinetic data. – *Biochem. J.*, 1986, **234**, 1, 21–29.
13. Cornish-Bowden, A., Porter, W. R., Trager, W. F. Evaluation of distribution-free confidence limits for enzyme kinetic parameters. – *J. Theor. Biol.*, 1978, **74**, 1, 163–175.
14. Crabbe, M. J. An enzyme kinetics program for desk-top computers. – *Comput. Biol. Med.*, 1982, **12**, 4, 263–283.
15. Crabbe, M. J. Distribution-free computer methods for analyzing ligand binding and enzyme mechanisms. – *Comput. Biol. Med.*, 1985, **15**, 3, 111–121.
16. Currie, D. J. Estimating Michaelis–Menten parameters: Bias, variance and experimental design. – *Biometrics*, 1982, **38**, 4, 907–919.
17. Dowd, J. E., Riggs, D. S. A comparison of estimates of Michaelis–Menten kinetic constants from various linear transformations. – *J. Biol. Chem.*, 1965, **240**, 2, 863–869.
18. Duggleby, R. A non-linear regression program for small computers. – *Anal. Biochem.*, 1981, **110**, 1, 9–18.
19. Greco, W. R., Priorer, R. I., Sharma, M., Korytnyk, W. ROSFIT: An enzyme kinetics nonlinear regression curve fitting package for microcomputer. – *Comput. Biomed. Res.*, 1982, **15**, 1, 39–45.
20. Hanson, K. R., Lung, R., Havir, E. A computer program for fitting data to the Michaelis–Menten equation. – *Biochem. Biophys. Res. Commun.*, 1967, **29**, 2, 194–197.
21. Hoy, T. G., Goldberg, D. M. An ALGOL-program for the computer-assisted calculation of enzyme kinetic data. – *Int. J. Biomed. Comput.*, 1971, **2**, 1, 71–77.
22. Mannervik, B., Jakobson, I., Warholm, M. A new procedure to derive weighting factors for nonlinear regression analysis applied to enzyme kinetic data. – *Biochim. Biophys. Acta*, 1979, **567**, 1, 43–48.
23. Munson, P. J., Rodbard, D. LIGAND: A versatile computerized approach for characterization of ligand-binding systems. – *Anal. Biochem.*, 1980, **107**, 1, 220–239.
24. Oestreicher, E. G., Pinto, G. F. Pocket computer program for fitting Michaelis–Menten equation. – *Comput. Biol. Med.*, 1983, **13**, 2, 309–317.
25. Oestreicher, E. G., Pinto, G. F. A microcomputer program for fitting enzyme inhibition rate equations. – *Comput. Biol. Med.*, 1987, **17**, 1, 53–68.
26. Pereira, L. M., Maya, M. T., Morais, J. A. Application of non-parametric methods to pharmacokinetic models of the Michaelis–Menten type. – *Eur. J. Drug. Metab. Pharmacokinet.*, 1991, Spec. No 3, 497–500.
27. Pinto, G. F., Oestreicher, E. G. Pocket computer program for fitting the Hill equation. – *Comput. Biol. Med.*, 1984, **14**, 4, 507–511.
28. Pinto, G. F., Oestreicher, E. G. A microcomputer program for fitting two-substrate enzyme rate equations. – *Comput. Biol. Med.*, 1988, **18**, 2, 135–144.

29. Porter, W. R., Trager, W. F. Improved non-parametric statistical methods for the estimation of Michaelis–Menten kinetics parameters by the direct linear plot. – *Biochem. J.*, 1977, **161**, 2, 293–302.
30. Raaijmakers, J. G. W. Statistical analysis of the Michaelis–Menten equation. – *Biometrics*, 1987, **43**, 793–805.
31. Reich, J. G. Parameter estimation and enzyme kinetic models. – *FEBS Letters*, 1970, **9**, 5, 245–251.
32. Schremmer, S. D., Waser, M. R., Kohn, M. C., Garfinkel, D. A. A computer program for analyzing enzyme kinetics data using graphical display and statistical analysis. – *Comput. Biomed. Res.*, 1984, **17**, 3, 289–298.
33. Siano, D. B., Zyrskind, J. W., Fromm, H. J. A computer program for fitting and statistically analyzing initial rate data applied to bovine hexokinase type II isoenzyme. – *Arch. Biochem. Biophys.*, 1975, **170**, 2, 587–600.
34. Storer, A. C., Darlison, M. G., Cornish-Bowden, A. The nature of experimental error in enzyme kinetic measurements. – *Biochem. J.*, 1975, **151**, 2, 361–367.
35. Szigetti, L., Tanner, R. D. An error estimation of Michaelis–Menten (Monod)-type kinetics. – *Appl. Microbiol. Biotechnol.*, 1993, **38**, 15, 610–614.
36. Vega-Catalan, F. J. A method for conforming the pH dependence of the Michaelis parameters of nonallosteric enzymes to four kinetic schemes. – *Comput. Biomed. Res.*, 1990, **23**, 5, 447–454.
37. Wilkinson, G. N. Statistical estimations in enzyme kinetics. – *Biochem. J.*, 1961, **80**, 2, 324–332.

ARVUTIPROGRAMM ENSÜMIREAKTSIOONIDE KINEETIKA ANALÜÜSIKS MAKSIMAALSE TÕEPÄRA MEETODIL

Ülo VAHER

On esitatud arvutiprogramm parameetrite ja nende usalduspiiride arvutamiseks maksimaalse tõepära meetodi abil hüperboolsete kineetiliste kõverate analüüsil. Kõverad kirjeldavad Michaelise–Menteni tüüpi ensümaatilisi reaktsioone, nende inhibeerimist ja ligandite sidumist. On toodud programmi aluseks oleva meetodi teoreetiline matemaatilis-statistiline põhjendus ja arvutusvalemid. Programm on tarbijasõbralik, sisaldab menüüsid ja selgitusi kasutajale. Tulemused väljastatakse nii tabelina kui ka graafikuna. Programmi peaeliseks on see, et ta ei vaja parameetrite hindamiseks lähteväärtuste sisestamist ning teda ei häiri koonduvusprobleemid erinevalt seni publitseeritud programmidest, mis kasutavad mittelineaarset regressiooni vähimruutude meetodil. Seda näitavad ka Monte Carlo meetodil tehtud suuremahulised arvuti-eksperimentid programmi võrdlemiseks nii vähimruutude kui ka mitteparameetrisel meetodil baseeruvate programmidega. Programm on lühike, lihtne, töökindel ja kirjutatud üldlevinud GWBASIC keeles.

КОМПЬЮТЕРНАЯ ПРОГРАММА ДЛЯ ВЫЧИСЛЕНИЯ ПАРАМЕТРОВ КИНЕТИЧЕСКИХ КРИВЫХ ТИПА МИХАЭЛИСА-МЕНТЕНА МЕТОДОМ МАКСИМАЛЬНОГО ПРАВДОПОДОБИЯ

Юло ВАХЕР

Разработана компьютерная программа для вычисления параметров и их доверительных границ кинетических кривых типа Михаэлиса-Ментена, в том числе кривых ингибирования ферментов и связывания лигандов методом максимального правдоподобия. Дано математико-статистическое обоснование метода и приведены вычислительные формулы. В отличие от программ, работающих на основе метода наименьших квадратов, составленная программа не нуждается в предварительных оценках неизвестных параметров и не подвержена опасности потери сходимости. Программа работает интерактивно и имеет меню-управляемый интерфейс пользователя. Программа написана на широко распространенном языке GWBASIC и может запускаться на любом компьютере.

Результаты массовых компьютерных экспериментов методом Монте-Карло по проверке теоретических положений метода подтвердили его преимущества перед другими наилучшими и известными программами.